

**UNCLASSIFIED**

**AD 4 2 0 6 8 7**

**DEFENSE DOCUMENTATION CENTER**

**FOR**

**SCIENTIFIC AND TECHNICAL INFORMATION**

**CAMERON STATION, ALEXANDRIA, VIRGINIA**



**UNCLASSIFIED**

NOTICE: When government or other drawings, specifications or other data are used for any purpose other than in connection with a definitely related government procurement operation, the U. S. Government thereby incurs no responsibility, nor any obligation whatsoever; and the fact that the Government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data is not to be regarded by implication or otherwise as in any manner licensing the holder or any other person or corporation, or conveying any rights or permission to manufacture, use or sell any patented invention that may in any way be related thereto.

MASSACHUSETTS INSTITUTE OF TECHNOLOGY  
LINCOLN LABORATORY

43 G-1

A MULTI-CAVITY KLYSTRON COMPUTER PROGRAM  
INCLUDING POTENTIAL DEPRESSION CORRECTION

S. F. Catalano  
C. Guay  
G. L. Guernsey  
R. L. Pease  
M. E. Schwarz, Jr.

29 August 1963

The work reported in this document was performed at Lincoln Laboratory, a center for research operated by Massachusetts Institute of Technology. This document is Lincoln Laboratory Report 442, which is approved under U.S. Government sponsorship.

A MULTI-CAVITY KLYSTRON COMPUTER PROGRAM  
INCLUDING POTENTIAL DEPRESSION CORRECTION

ABSTRACT

This report describes a computer program which may be used to investigate the characteristics of a multicavity klystron<sup>1, 2, 3</sup>. It is written for the IBM 7090 in "Fortran" language so that it may be intelligible to people other than the originator of the program. It is written for solid or hollow beam klystrons. For maximum flexibility it is in the form of several subroutines connected by a master control program. It is complete in the sense that it leaves nothing for the user to compute manually.

S. F. Catalano\*

C. Guay<sup>#</sup>

G. L. Guernsey


R. L. Pease<sup>+</sup>

M. E. Schwarz, Jr.

\* Group 45

<sup>#</sup> Now at Antenna Systems Inc., Manchester, N. H.

<sup>+</sup> Now Assoc. Prof., Brooklyn College, Brooklyn, N. Y.



## TABLE OF CONTENTS

I.	INTRODUCTION	1
II.	INPUT	2
III.	OUTPUT	6
IV.	THE MECHANICS OF USING THE PROGRAM	11
	APPENDIX: FORTRAN PROGRAM LISTING	
	Master Control Program	14
	Subroutines	
	POTDEP	23
	SPCRDP	25
	SPCRDM	26
	EFCMH	27
	FREQ	28
	TERPOL	28
	AINTER	28
	FOMATR	29
	DMATRX	30
	CMMP	31
	CMSUBT	31
	RMMP	31
	REFERENCES	32

## I. Introduction

The following describes a multicavity klystron (MCK) computer program which has been written for the IBM 7090 computer at Lincoln Laboratory. It is based upon two papers<sup>1,2</sup> describing an improved MCK analysis and a third<sup>3</sup> in which a potential depression correction is derived.

It has been assumed that the reader is familiar with the above mentioned reports, and this manual is intended only as an aid to the use of the program.

The program was written in the IBM 709 'FORTRAN' language which, in addition to being a compact and efficient language, provides an excellent record of the computation which is easily understood by people other than the originator of the program. The writers have sought to achieve three qualities in the program: generality, adaptability to future changes, and completeness. The program achieves generality insofar as it is good for any solid or hollow beam klystron with from 1 to 6 (this number can be extended by altering input and output format statements) cavities of equal or unequal spacing.

The program is flexible and adaptable to future changes since it has been written in the form of several subroutines connected by a master control program. Any future change will merely require altering or adding subroutines without disturbing the rest of the program. \*

The program is complete in the sense that it leaves nothing for the user to compute manually. This can be seen from the following outline in which only the most elementary parameters of the tube are required as input to the program.

---

\*No attempt has been made to optimize the program for the sake of brevity or economy of operation, nor is such optimization planned. The writers realize that several correctable redundancies in form do exist, and that some inefficient methods are employed. Users of the program may find it to their advantage to make several minor alterations.

## II. Input

The input data are of two kinds. First, there are those which put into storage the mechanical and electrical tube parameters and associated quantities necessary to the computation. Table I lists and defines these parameters. Secondly, there are those inputs, listed in Table II, which determine the quality and quantity of the output data. Some discussion of these is necessary.

The quality of the output is determined to some degree by the number of modes considered, although it has been found for typical tubes that the effect due to modes of order higher than the fourth contribute little to the end results. The decision of the degree of accuracy desired has been left to the user in that the number of modes to be used in the calculation is an input parameter.

Since the output is generally in the form of some parameter as a function of frequency, the number of frequencies at which data is desired partially determines the amount of output data. Output data may be obtained over a 16% or smaller bandwidth for as many frequencies as are desired. The input necessary to determine the bandwidth and number of frequency points is: the fractional deviation from the center frequency of the initial (lowest) operating frequency at which data is desired, the fractional frequency increment, and the total number of points desired.

There are, in addition, three other parameters not directly used in the computation. These will have value 1 or 0 and will determine which of several options in output are to be chosen.

It might be advantageous for a particular tube to change the values of  $\gamma_{\pm}$  at which the searches for solutions start and the spacing of the points used in the search. (Note  $\gamma_{\pm} = (\beta_{\pm}^2 - k^2)^{1/2}$ , where  $\beta_{\pm}$  are the roots of the propagation equation and  $k \equiv \omega/c$ .) If the range is too wide or the initial spacing too fine, computer time is wasted; if the range is too narrow or the initial spacing

too coarse, a root may be missed. As the program is presently set up, the search for  $\gamma_+$  starts at  $0.6\beta_e$  and increases, and the search for  $\gamma_-$  starts at  $1.8\beta_e$  and decreases. This change may be effected by altering statement number 121 and the immediately following (unnumbered) statement which reads presently:

```

121          GPOS(I) = 0.6*BE(I)
          GMINUS(I) = 1.8*BE(I)

```

(Note  $\beta_e = \omega/v$ .) This is adequate for conventional multicavity klystrons where  $v/c$  is of the order of 0.5 and where the  $\beta_{\pm}$  are spaced on the order of  $\pm 0.1$  from  $\beta_e$  (i.e.,  $\omega_q/\omega \approx 0.1$ ). Similarly, the initial spacing of the points used to search for the root can be changed by altering statement number 153 which reads presently:

```

153          DDG = .001

```

Increasing the value of "DDG" widens the initial spacing.

## INPUT

TABLE I

## TUBE PARAMETERS

<u>Fortran Symbol</u>	<u>Parameter</u>	<u>Definition</u>
NCVS	$r$	Number of cavities
VO	$V_o$	DC voltage
FIO	$I_o$	DC current
FO	$f_o$	Center frequency
A	$a$	Drift tube inside radius (inches)
B	$b$	*Beam outside radius (inches)
C	$c$	*Beam inside radius (inches)
DGAP	$d$	Gap half length (inches)
RQ(I)	$(R/Q)^{(i)}$	(R/Q) value for each cavity
QO(I)	$Q_o^{(i)}$	Circuit loading for each cavity
QEXT(I)	$Q_{ext}^{(i)}$	External loading for each cavity
S(I)	$l^{(i+1)} - l^{(i)}$	Center gap to center gap distance for each gap pair
THETA	$\theta_{ext}$	Phase shift in external circuits
FNU(I)	$\nu^{(i)}$	Fractional detuning for each cavity

\*In this treatment, infinite magnetic field is assumed, so that there is no variation of beam dimensions.

TABLE II  
OUTPUT OPTIONS

<u>Fortran Symbol</u>	<u>Parameter</u>	<u>Definition</u>
N	n	Number of modes
SFRIQ	$\delta_o = \frac{f-f_o}{f_o}$	Fractional deviation of input frequency (f) from center frequency (f <sub>o</sub> ) (SFRIQ $\geq$ - .08)
FFDEL	$\frac{\Delta f}{f_o}$	Fractional frequency increment
NKK		Number of frequencies at which information is desired (Note: SFRIQ + NKK * FFDEL must be less than or equal to + .08)
POTD		= 0 do potential depression ≠ 0 omit potential depression
SKIP		= 0 do complete computation thru gain and phase as a function of frequency ≠ 0 do all calculations except gain and phase as a function of frequency
WOT		≠ 0 print all output parameters = 0 print only gain and phase information

### III. Output

The heading or the first nineteen lines of the printed output which are always printed verify some of the input parameters and serve to identify the tube and conditions being examined. Each of these lines has a label. The contents of these lines are:

1. The contents of identification input card number 1
2. Number of cavities
3. Number of modes
4. DC voltage
5. DC current
6. Center frequency
7. A -
8. B -
9. C -

Lines 10 through 16 give the following parameters for each of the cavities considered and the drift tube.

10. Cavity numbers<sup>\*</sup>
11. DC cavity voltage (depressed potential)
12. D
13. R/Q
14. QO
15. QEXT
16. S

---

\* The information for the beam in the drift tube is presented with the cavity data in the last column (i. e., as cavity number NCVS + 1).

The heading is completed by printing out the three input parameters:

17. NKK
18. FREQ = (SFRIQ)
19. DEL = (FFDEL)

The second portion of the output is controlled by the WOT input option parameter. It will be printed out if WOT  $\neq$  0. It consists of major groupings for each frequency and minor groups for each mode at each particular frequency. The major group is denoted by the printed line:

"FDEL = 'xyz....' where 'xyz' is  $(\text{frequency} - f_o)/f_o$ . This will be followed by as many groups of the following twenty-eight parameters, as there are modes to be considered. Each minor group is headed by the line "M = x" where x is the mode number. Each parameter is given in tabular form for each cavity. The parameters again have a label. They are:

<u>Fortran Symbol</u>	<u>Parameter</u>	<u>Definition</u>
BE	$\beta_e$	
B+	$\beta_{+n}$	Fast wave longitudinal plasma wave number
B-	$\beta_{-n}$	Slow wave longitudinal plasma wave number
G+	$\gamma_{+n}$	Fast wave transverse E. M. wave number
G-	$\gamma_{-n}$	Slow wave transverse E. M. wave number
T+	$T_{+n}$	Fast wave transverse plasma wave number
T-	$T_{-n}$	Slow wave transverse plasma wave number
BQPOS	$\beta_{q+n}$	
BQMIN	$\beta_{q-n}$	

<u>Fortran Symbol</u>	<u>Parameter</u>
WQ+/WP	$\omega_{q+n}/\omega_p$
WQ-/WP	$\omega_{q-n}/\omega_p$
WQ+/W	$\omega_{q+n}/\omega$
WQ-/W	$\omega_{q-n}/\omega$
ZETA +	$\zeta_{+n}$
ZETA -	$\zeta_{-n}$
$Y\phi+ /G\phi$	$Y_{o+n}/G_o$
$Y\phi- /G\phi$	$Y_{o-n}/G_o$
E+	$E_{+n}$
E-	$E_{-n}$
F+	$F_{+n}$
F-	$F_{-n}$
C+	$C_{+n}$
C-	$C_{-n}$
M+	$M_{+n}$
M-	$M_{-n}$
H+	$H_{+n}$
H-	$H_{-n}$
GELGO	$(G_{el}/G_o)_n$

At the end of each frequency group, there are printed out (if WOT  $\neq$  0) the following:

<u>Fortran Symbol</u>	<u>Parameter</u>	<u>Definition</u>
GELGOP	$G_{el}/G_o = \sum_n (G_{el}/G_o)_n$	
QEL	$Q_{el}$	Q of electronic load
Q	Q	total Q
QOEL	$Q_{oel}$	Q of cavity and electronic load
GEL	$G_{el}$	electronic conductance
REL	$R_{el}$	$1/G_{el}$
if WOT $\neq$ 0 and SKIP = 0		
Y	$y^{(i)}$	$Q^{(i)}/Q_{el}^{(i)}$
ZGR	$\text{Re}(Z_g^{(i)})$	Real part of gap impedance
ZGI	$\text{Im}(Z_g^{(i)})$	Imaginary part of gap impedance
FIVR		Real part of over-all transadmittance
FIVI		Imaginary part of over-all transadmittance
GVR	$\text{Re}(G_v)$	Real part of over-all voltage gain
GVI	$\text{Im}(G_v)$	Imaginary part of over-all voltage gain

After all of the major (frequency) groupings are printed out, the cavity detunings are given. These are labeled NU and given for each cavity. Finally, the numerical gain, gain in db, and phase are given as a function of frequency. These quantities are unlabeled but appear in three distinct groups in which

frequency increases from left to right in steps according to the frequency increment chosen. These quantities (gain, phase) are calculated only if SKIP = 0.



#### IV. The Mechanics of Using the Program

The mechanics of using the program are very simple. They consist essentially of knowing how to get the input parameters into the computer. This is accomplished by punching the values of the input parameters on a set of cards in the format\* and sequence indicated below, placing these at the end of the program deck, and instructing the operator to clear the core and load the cards. The card format for input parameters is as follows:

##### Card Format for Input Parameters

###### Card 1

Column 1 requires a "1" punch if it is desired to skip to a new page for each tube. Columns 2-72 may contain any identifying information for the printed output such as tube name and date.

###### Card 2

Parameter	N	NCVS	VO	FIO	FO	A	B	C
Columns	1-5	6-10	11-18	19-26	27-34	35-41	42-48	49-55
Type Conversion	I	I	E	E	E	F	F	F
Decimal Places			0	0	0	4	4	4

###### Card 3

Parameter	NKK	SFRIQ	FFDEL
Columns	1-5	6-15	16-25
Type Conversion	I	F	F
Decimal Places		4	4

---

\* See FORTRAN manual for an explanation of conversion types.

Card 4

Parameter	DGAP(1)	DGAP(2)	DGAP(3)	DGAP(4)	DGAP(5)	DGAP(6)
Columns	1-8	9-16	17-24	25-32	33-40	41-48
Type Conversion	F	F	F	F	F	F
Decimal Places	4	4	4	4	4	4

Card 5

Parameter	RQ(1)	RQ(2)	RQ(3)	RQ(4)	RQ(5)	RQ(6)
Columns	1-10	11-20	21-30	31-40	41-50	51-60
Type Conversion	F	F	F	F	F	F
Decimal Places	2	2	2	2	2	2

Card 6

Parameter	QO(1)	QO(2)	QO(3)	QO(4)	QO(5)	QO(6)
Columns	1-10	11-20	21-30	31-40	41-50	51-60
Type Conversion	F	F	F	F	F	F
Decimal Places	2	2	2	2	2	2

Card 7

Parameter	QEXT(1)	QEXT(2)	QEXT(3)	QEXT(4)	QEXT(5)	QEXT(6)
Columns	1-10	11-20	21-30	31-40	41-50	51-60
Type Conversion	F	F	F	F	F	F
Decimal Places	2	2	2	2	2	2

Card 8

Parameter	S(1)	S(2)	S(3)	S(4)	S(5)
Columns	1-8	9-16	17-24	25-32	33-40
Type Conversion	F	F	F	F	F
Decimal Places	4	4	4	4	4

#### Card 9

Parameter	THETA
Columns	1 - 10
Type Conversion	F
Decimal Places	4

#### Card 10

Parameter	POTD	WOT	SKIP
Columns	1 - 10	11 - 20	21 - 30
Type Conversion	F	F	F
Decimal Places	2	2	2

The input should end with card 10 if tunings are not to be included. If there is another tube to be analyzed, the next card will be of the same format as Card 1 above, etc. The program stops when it runs out of cards. In this case, the value of SKIP must be nonzero. Cards 11 and 12 are to be included if, and only if, SKIP = 0.

#### Card 11

Parameter	FNU(1)	FNU(2)	FNU(3)	FNU(4)	FNU(5)	FNU(6)
Columns	1 - 10	11 - 20	21 - 30	31 - 40	41 - 50	51 - 60
Type Conversion	F	F	F	F	F	F
Decimal Places	5	5	5	5	5	5

This card may be followed by as many cards of the same format as there are sets of tunings to be investigated.

#### Card 12

Punching "999.99999" in columns 1 - 10 indicates to the program that there are no more sets of tunings to be investigated for this particular tube and to proceed to the next tube.

# MASTER CONTROL PROGRAM

C  
C

## REVISED MCK PROGRAM PART 1

```

    DIMENSION DBGAIN(200),REL(10)
    DIMENSION DGAP(10),RQ(10),QO(10),QEXT(10),S(10),VI(10),VOC(10),VS(
110),RHO(10),BP(10),VBAR(10),BE(10),FK(10),BQPOS(10,10),WQWPP(10),W
20WP(10),GPOS(10),BPOS(10),TPOS(10),ZETAP(10,10),YOGOP(10,10),BMINU
3S(10),GMINUS(10),TMINUS(10),BQMIN(10,10),WQWPM(10),WQWM(10),ZETAM(
410,10),YOGOM(10,10),EPOS(10),FPOS(10),CKP(10,10),FMP(10,10),FHP(10
5,10),EMIN(10),FMIN(10),CKM(10,10),FMM(10,10),FHM(10,10),GELGO(10),
6GELGOP(10),QEL(10),G(10),GO(10),GP(10),GM(10),D(10),QOEL(10),GPST(
710),GMST(10),GAIN(200),PHASE(200),VPTR(10),VSUBD(10),FFO(200)
    DIMENSION FNU(10),FREAL(10,10),FIMAG(10,10),Y(10),GEL(10),ZGR(10),
12GI(10),PPLUSR(20,20),PPLUSI(20,20),COLMR(20,20),COLMI(20,20),FLMP
2R(20,20),FLMPI(20,20),DR(20,20),DI(20,20),FIR(20,20),FII(20,20),FM
3ULTR(20,20),FMULTI(20,20),SUMR(20,20),SUMI(20,20),CR(20,20),CI(20,
420),ZGGLR(20,20),ZGGLI(20,20)
    DIMENSION BEE(3,10,10),FKKKK(3,10,10),GPOSS(3,10,10),BPOSS(3,10,10
1),TPOSS(3,10,10),BQPOSS(3,10,10),WQWPPP(3,10,10),WQWPW(3,10,10),ZE
2TAPP(3,10,10),YOGOPP(3,10,10),GMINSS(3,10,10),BMINSS(3,10,10),TMIN
3SS(3,10,10),BQMINN(3,10,10),WQWPMM(3,10,10),WQWMW(3,10,10),ZETAMM(
43,10,10),YOGOMM(3,10,10),GELGPP(3,10,10),EPOSS(3,10,10),FPOSS(3,10
5,10),CKPP(3,10,10),FMPP(3,10,10),FHPP(3,10,10),EMINN(3,10,10),FMIN
6N(3,10,10),CKMM(3,10,10),FMMM(3,10,10),FHMM(3,10,10),GELGOO(3,10,1
70)
    COMMON BEE,FKKKK,GPOSS,BPOSS,TPOSS,BQPOSS,WQWPPP,WQWPW,ZETAPP,YOGO
1PP,GMINSS,BMINSS,TMINSS,BQMINN,WQWPMM,WQWMW,ZETAMM,YOGOMM,GELGPP,E
2POSS,FPOSS,CKPP,FMPP,FHPP,EMINN,FMINN,CKMM,FMMM,FHMM,GELGOO
101  READ INPUT TAPE 2,1
    READ INPUT TAPE 2,2,N,NCVS,VO,FIO,FO,A,B,C
    READ INPUT TAPE 2,9,NKK,SFRIQ,FFDEL
    READ INPUT TAPE 2,3,(DGAP(I),I=1,NCVS)
    READ INPUT TAPE 2,4,(RQ(I),I=1,NCVS)
    READ INPUT TAPE 2,4,(QO(I),I=1,NCVS)
    READ INPUT TAPE 2,4,(QEXT(I),I=1,NCVS)
    FRIQ=SFRIQ
    NK=3
    FREQM=-0.0800
    FDEL=0.0800
    NCVSL1=NCVS-1
    NCVSP1=NCVS+1
    READ INPUT TAPE 2,3,(S(I),I=1,NCVSL1)
    READ INPUT TAPE 2,5,THETA
    READ INPUT TAPE 2,20,POTD,WOT,SKIP
    EM=1.759E11
    CL=2.998E8
    EMC2=EM/CL**2
    THETA=THETA*3.14159265/180.0
    EPSO=8.854E-12
    PI=3.14159265
    IF (POTD) 106,107,106
107  CALL POTDEP(NCVS,VO,FIO,A,B,C,DGAP,VPTR,RSUBD,VBAR,VSUBD)
    DO 110 I=1,NCVS
    VI(I)=VBAR(I)
    GO(I)=FIO/VI(I)
110  CONTINUE
    VI(NCVSP1)=VSUBD(NCVSP1)

```

N.B.: The Bessel function subroutine called for in this program as BESSELF is GEBSL, Share distribution number 271.

```

      GO(NCVSP1)=FIO/VI(NCVSP1)
      GO TO 111
106  DO 104 I=1,NCVSP1
      VI(I)=VO
104  GO(I)=FIO/VI(I)
111  WRITE OUTPUT TAPE 3,1
      WRITE OUTPUT TAPE 3,6,NCVS,N,VO
      WRITE OUTPUT TAPE 3,14,FIO,F0,A,B,C
      WRITE OUTPUT TAPE 3,13
      WRITE OUTPUT TAPE 3,12,(VI(I),I=1,NCVSP1)
      WRITE OUTPUT TAPE 3,15,(DGAP(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,16,(RQ(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,17,(QO(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,18,(QEXT(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,19,(S(I),I=1,NCVSL1)
      WRITE OUTPUT TAPE 3,56,NKK,FRIQ,FFDEL
      IF(NKK-1) 320,321,322
320  PAUSE
321  NK=1.00
      FREQM=0.00
322  DO 103 K=1,NK
      FFO(K)=FREQM*F0+F0
      DO 115 I=1,NCVSP1
      VOC(I)=SQRTF(1.0-1.0/(1.0+EMC2*VI(I))**2)
      VS(I)=VOC(I)*CL
      RHO(I)=(39.37**2)*FIO/(PI*(B**2-C**2)*VS(I))
      BP(I)=SQRTF(EM*RHO(I)/EPSO*SQRTF((1.0-VOC(I)**2)**3))/(VS(I)*39.37
1)
      BE(I)=2.0*PI*FFO(K)/(VS(I)*39.37)
      FK(I)=BE(I)*VOC(I)
      IF (K-1) 120,121,122
120  PAUSE
121  GPOS(I)=0.6*BE(I)
      GMINUS(I)=1.8*BE(I)
      GO TO 115
122  GPOS(I)=0.97*GPST(I)
      GMINUS(I)=2.30*GMST(I)
115  CONTINUE
128  DO 105 L=1,N
      DO 102 I=1,NCVSP1
      IF (I-1) 160,161,162
160  PAUSE
162  III=I-1
      DO 163 II=1,III
      IF (VI(I)-VI(II)) 163,164,163
163  CONTINUE
161  GP(I)=GPOS(I)
      GM(I)=GMINUS(I)
153  DDG=0.001
154  CALL SPCRDP(I,A,B,C,DDG,GP,BP,BE,FK,GGPOS,BBPOS,TTPOS)
      GPOS(I)=GGPOS
      BPOS(I)=BBPOS
      TPOS(I)=TTPOS
      BQPOS(L,I)=BE(I)-BPOS(I)
      WQWPP(I)=BQPOS(L,I)/BP(I)
      WQWP(I)=BQPOS(L,I)/BE(I)
      ZETAP(L,I)=BPOS(I)*BQPOS(L,I)*TPOS(I)**2/(GPOS(I)**2*(GPOS(I)**2+T
1POS(I)**2))

```

```

YOGOP(L,I)=BE(I)/(BQPOS(L,I)*(1.0+EMC2*VI(I))*(2.0+EMC2*VI(I)))
CALL SPCRDM(I,A,B,C,DDG,GM,BP,BE,FK,GGMIN,BBMIN,TTMIN)
GMINUS(I)=GGMIN
BMINUS(I)=BBMIN
TMINUS(I)=TTMIN
BQMIN(L,I)=-(BE(I)-BMINUS(I))
WQWPM(I)=BQMIN(L,I)/BP(I)
WQWM(I)=BQMIN(L,I)/BE(I)
ZETAM(L,I)=BMINUS(I)*BQMIN(L,I)*TMINUS(I)**2/(GMINUS(I)**2*(GMINUS
1(I)**2+TMINUS(I)**2))
YOGOM(L,I)=BE(I)/(BQMIN(L,I)*(1.0+EMC2*VI(I))*(2.0+EMC2*VI(I)))
GO TO 181
164 GPOS(I)=GPOS(II)
BPOS(I)=BPOS(II)
TPOS(I)=TPOS(II)
BQPOS(L,I)=BQPOS(L,II)
WQWPP(I)=WQWPP(II)
WQWP(I)=WQWP(II)
ZETAP(L,I)=ZETAP(L,II)
YOGOP(L,I)=YOGOP(L,II)
GMINUS(I)=GMINUS(II)
BMINUS(I)=BMINUS(II)
TMINUS(I)=TMINUS(II)
BQMIN(L,I)=BQMIN(L,II)
WQWPM(I)=WQWPM(II)
WQWM(I)=WQWM(II)
ZETAM(L,I)=ZETAM(L,II)
YOGOM(L,I)=YOGOM(L,II)
181 BEE(K,L,I)=BE(I)
FKKKK(K,L,I)=FK(I)
GPOSS(K,L,I)=GPOS(I)
RPOSS(K,L,I)=BPCS(I)
TPOSS(K,L,I)=TPOS(I)
BQPOSS(K,L,I)=BQPOS(L,I)
WQWPPP(K,L,I)=WQWPP(I)
WQWPW(K,L,I)=WQWP(I)
ZETAPP(K,L,I)=ZETAP(L,I)
YOGOPP(K,L,I)=YOGOP(L,I)
GMINSS(K,L,I)=GMINUS(I)
BMINSS(K,L,I)=BMINUS(I)
TMINSS(K,L,I)=TMINUS(I)
BQMINN(K,L,I)=BQMIN(L,I)
WQWPMM(K,L,I)=WQWPM(I)
WQWMW(K,L,I)=WQWM(I)
ZETAMM(K,L,I)=ZETAM(L,I)
YOGOMM(K,L,I)=YOGOM(L,I)
IF (L-1) 125,126,102
125 PAUSE
126 GPST(I)=GPOS(I)
GMST(I)=GMINUS(I)
102 CONTINUE
CALL EFCMH(L,A,B,C,NCVS,VI,TPOS,GPOS,BPOS,EPOS,FPOS,CKP,DUM1,DUM2,
1DUM3)
CALL EFCMH(L,A,B,C,NCVS,VI,TMINUS,GMINUS,BMINUS,EMIN,FMIN,CKM,DUM1
1,DUM2,DUM3)
DO 108 I=1,NCVS
FMP(L,I)=BESSELF(DGAP(I)*BPOS(I),0,1)
FMM(L,I)=BESSELF(DGAP(I)*BMINUS(I),0,1)

```

```

FHP(L,I)=EPOS(I)+(ZETAP(L,I)*FPOS(I))
FHM(L,I)=EMIN(I)-(ZETAM(L,I)*FMIN(I))
GELGO(I)=0.25*(FHP(L,I)*YOGOP(L,I)*CKP(L,I)**2*FMP(L,I)**2/((1.0+Z
1ETAP(L,I))**2)-(FHM(L,I)*YOGOM(L,I)*CKM(L,I)**2*FMM(L,I)**2/((1.0-
2ZETAM(L,I))**2)))
GELGOO(K,L,I)=GELGO(I)
EPOSS(K,L,I)=EPOS(I)
FPOSS(K,L,I)=FPOS(I)
CKPP(K,L,I)=CKP(L,I)
FMPP(K,L,I)=FMP(L,I)
FHPP(K,L,I)=FHP(L,I)
EMINN(K,L,I)=EMIN(I)
FMINN(K,L,I)=FMIN(I)
CKMM(K,L,I)=CKM(L,I)
FMMM(K,L,I)=FMM(L,I)
FHMM(K,L,I)=FHM(L,I)
108 CONTINUE
105 CONTINUE
FREQM=FREQM+FDEL
103 CONTINUE
MM=0
IF(SKIP) 176,175,176
175 READ INPUT TAPE 2,49,(FNU(I),I=1,NCVS)
IF(FNU(1)-999.999) 400,101,101
400 FRIQ=SFRIQ
176 DO 171 K=1,NKK
FFO(K)=FRIQ*FO+FO
DO 189 I=1,NCVS
GELGOP(I)=0.0
189 CONTINUE
IF(WOT) 179,180,179
179 IF(MM) 180,401,180
401 WRITE OUTPUT TAPE 3,10,FRIQ
180 DO 172 L=1,N
IF(NKK-1) 323,324,325
323 PAUSE
325 CALL TERPOL(NCVSP1,L,BEE,FRIQ,BE)
CALL TERPOL(NCVSP1,L,BEE,FRIQ,BE)
CALL TERPOL(NCVSP1,L,FKKKK,FRIQ,FK)
CALL TERPOL(NCVSP1,L,GPOSS,FRIQ,GPOS)
CALL TERPOL(NCVSP1,L,BPOSS,FRIQ,BPOS)
CALL TERPOL(NCVSP1,L,TPOSS,FRIQ,TPOS)
CALL TERPOL(NCVSP1,L,WQWPW,FRIQ,WQWP)
CALL TERPOL(NCVSP1,L,WQWMW,FRIQ,WQWM)
CALL TERPOL(NCVSP1,L,WQWPP,FRIQ,WQWP)
CALL TERPOL(NCVSP1,L,GMINSS,FRIQ,GMINUS)
CALL TERPOL(NCVSP1,L,BMINSS,FRIQ,BMINUS)
CALL TERPOL(NCVSP1,L,TMINSS,FRIQ,TMINUS)
CALL TERPOL(NCVSP1,L,WQWPM,FRIQ,WQWP)
CALL AINTER(NCVSP1,L,BQPOSS,FRIQ,BQPOS)
CALL AINTER(NCVSP1,L,ZETAPP,FRIQ,ZETAP)
CALL AINTER(NCVSP1,L,YOGOPP,FRIQ,YOGOP)
CALL AINTER(NCVSP1,L,BQMINN,FRIQ,BQMIN)
CALL AINTER(NCVSP1,L,ZETAMM,FRIQ,ZETAM)
CALL AINTER(NCVSP1,L,YOGOMM,FRIQ,YOGOM)
CALL TERPOL(NCVS,L,GELGOO,FRIQ,GELGO)

```

```

CALL TERPOL(NCVS,L,EPOSS,FRIQ,EPOS)
CALL TERPOL(NCVS,L,FPOSS,FRIQ,FPOS)
CALL TERPOL(NCVS,L,EMINN,FRIQ,EMIN)
CALL TERPOL(NCVS,L,FMINN,FRIQ,FMIN)
CALL AINTER(NCVS,L,CKPP,FRIQ,CKP)
CALL AINTER(NCVS,L,FMPP,FRIQ,FMP)
CALL AINTER(NCVS,L,FHPP,FRIQ,FHP)
CALL AINTER(NCVS,L,CKMM,FRIQ,CKM)
CALL AINTER(NCVS,L,FMMM,FRIQ,FMM)
CALL AINTER(NCVS,L,FHMM,FRIQ,FHM)
GO TO 351
324 DO 349 I=1,NCVSP1
    BE(I)=BEE(1,L,I)
    FK(I)=FKKKK(1,L,I)
    GPOS(I)=GPOSS(1,L,I)
    BPOS(I)=BPOSS(1,L,I)
    TPOS(I)=TPOSS(1,L,I)
    WQWP(I)=WQWPW(1,L,I)
    WQWM(I)=WQWMW(1,L,I)
    WQWPP(I)=WQWPPP(1,L,I)
    GMINUS(I)=GMINSS(1,L,I)
    BMINUS(I)=BMINSS(1,L,I)
    TMINUS(I)=TMINSS(1,L,I)
    WQWPM(I)=WQWPMM(1,L,I)
    BQPOS(L,I)=BQPOSS(1,L,I)
    ZETAP(L,I)=ZETAPP(1,L,I)
    YOGOP(L,I)=YOGOPP(1,L,I)
    BQMIN(L,I)=BQMINN(1,L,I)
    ZETAM(L,I)=ZETAMM(1,L,I)
    YOGOM(L,I)=YOGOMM(1,L,I)
349 CONTINUE
    DO 350 I=1,NCVS
        GELGO(I)=GELGOO(1,L,I)
        EPOS(I)=EPOSS(1,L,I)
        FPOS(I)=FPOSS(1,L,I)
        EMIN(I)=EMINN(1,L,I)
        FMIN(I)=FMINN(1,L,I)
        CKP(L,I)=CKPP(1,L,I)
        FMP(L,I)=FMPP(1,L,I)
        FHP(L,I)=FHPP(1,L,I)
        CKM(L,I)=CKMM(1,L,I)
        FMM(L,I)=FMMM(1,L,I)
        FHM(L,I)=FHMM(1,L,I)
350 CONTINUE
351 DO 183 I=1,NCVS
    GELGOP(I)=GELGOP(I)+GELGO(I)
183 CONTINUE
    IF (WOT) 178,172,178
178 IF(MM) 172,402,172
402 WRITE OUTPUT TAPE 3,11,L
    WRITE OUTPUT TAPE 3,26,(BE(I),I=1,NCVSP1)
    WRITE OUTPUT TAPE 3,7,(BPOS(I),I=1,NCVSP1)
    WRITE OUTPUT TAPE 3,8,(BMINUS(I),I=1,NCVSP1)
    WRITE OUTPUT TAPE 3,21,(GPOS(I),I=1,NCVSP1)
    WRITE OUTPUT TAPE 3,22,(GMINUS(I),I=1,NCVSP1)
    WRITE OUTPUT TAPE 3,23,(TPOS(I),I=1,NCVSP1)
    WRITE OUTPUT TAPE 3,24,(TMINUS(I),I=1,NCVSP1)
    WRITE OUTPUT TAPE 3,66,(BQPOS(L,I),I=1,NCVSP1)

```

```

WRITE OUTPUT TAPE 3,67,(BQMIN(L,I),I=1,NCVSP1)
WRITE OUTPUT TAPE 3,25,(WQWPP(I),I=1,NCVSP1)
WRITE OUTPUT TAPE 3,27,(WQWPM(I),I=1,NCVSP1)
WRITE OUTPUT TAPE 3,28,(WQWP(I),I=1,NCVSP1)
WRITE OUTPUT TAPE 3,29,(WQWM(I),I=1,NCVSP1)
WRITE OUTPUT TAPE 3,30,(ZETAP(L,I),I=1,NCVSP1)
WRITE OUTPUT TAPE 3,31,(ZETAM(L,I),I=1,NCVSP1)
WRITE OUTPUT TAPE 3,32,(YOGOP(L,I),I=1,NCVSP1)
WRITE OUTPUT TAPE 3,33,(YOGOM(L,I),I=1,NCVSP1)
WRITE OUTPUT TAPE 3,34,(EPOS(I),I=1,NCVS)
WRITE OUTPUT TAPE 3,35,(EMIN(I),I=1,NCVS)
WRITE OUTPUT TAPE 3,36,(FPOS(I),I=1,NCVS)
WRITE OUTPUT TAPE 3,37,(FMIN(I),I=1,NCVS)
WRITE OUTPUT TAPE 3,38,(CKP(L,I),I=1,NCVS)
WRITE OUTPUT TAPE 3,39,(CKM(L,I),I=1,NCVS)
WRITE OUTPUT TAPE 3,40,(FMP(L,I),I=1,NCVS)
WRITE OUTPUT TAPE 3,41,(FMM(L,I),I=1,NCVS)
WRITE OUTPUT TAPE 3,42,(FHP(L,I),I=1,NCVS)
WRITE OUTPUT TAPE 3,43,(FHM(L,I),I=1,NCVS)
WRITE OUTPUT TAPE 3,44,(GELGO(I),I=1,NCVS)
172 CONTINUE
   IF (WOT) 184,185,184
184   IF(MM) 185,403,185
403   WRITE OUTPUT TAPE 3,45,(GELGOP(I),I=1,NCVS)
      DO 118 I=1,NCVS
      QEL(I)=1.0/(RO(I)*GELGOP(I)*GO(I))
      Q(I)=1.0/(1.0/QO(I)+1.0/QEL(I)+1.0/QEXT(I))
      QOEL(I)=1.0/(1.0/Q(I)-1.0/QEXT(I))
      GEL(I)=GELGOP(I)*GO(I)
      REL(I)=1.0/GEL(I)
118   CONTINUE
      WRITE OUTPUT TAPE 3,46,(QEL(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,47,(Q(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,48,(QOEL(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,53,(GEL(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,81,(REL(I),I=1,NCVS)
185   IF(SKIP) 177,186,177
186   NN=2*N
      DELO=(FFO(K)-FO)/FO
      CALL FREQ(Q,FNU,DELO,NCVS,FREAL,FIMAG)
      DO 143 I=1,NCVS
      Y(I)=Q(I)/QEL(I)
      ZGR(I)=Y(I)*FREAL(I)/GEL(I)
      ZGI(I)=Y(I)*FIMAG(I)/GEL(I)
      CALL FOMATR(N,I,NCVS,CKP,CKM,FMP,FMM,FHP,FHM,ZETAP,ZETAM,YOGOP,YOG
10M,PPLUSR,PPLUSI,COLMR,COLMI,FLMPR,FLMPI)
140   CALL DMATRX(NCVSP1,NN,I,BQPOS,BQMIN,BE,S,DR,DI,FIR,FII)
      IF (I-1) 130,131,132
130   PAUSE
132   IF (I-NCVS) 133,134,130
131   CALL CMMP(DR,DI,NN,NN,COLMR,COLMI,1,FMULTR,FMULTI)
      GO TO 143
133   DO 170 JJ=1,NN
      DO 170 KK=1,NN
      ZGGLR(JJ,KK)=GO(I)*ZGR(I)*FLMPR(JJ,KK)-GO(I)*ZGI(I)*FLMPI(JJ,KK)
      ZGGLI(JJ,KK)=GO(I)*ZGI(I)*FLMPR(JJ,KK)+GO(I)*ZGR(I)*FLMPI(JJ,KK)
170   CONTINUE
      CALL CMSUBT(ZGGLR,FIR,NN,SUMR)

```

```

      CALL CMSUBT(ZGGLI,FII,NN,SUMI)
      CALL CMMP(SUMR,SUMI,NN,NN,FMULTR,FMULTI,1,CR,CI)
      DO 141 JJ=1,NN
      KK=1
      FMULTR(JJ,KK)=CR(JJ,KK)
      FMULTI(JJ,KK)=CI(JJ,KK)
141  CONTINUE
      CALL CMMP(DR,DI,NN,NN,FMULTR,FMULTI,1,CR,CI)
      DO 142 JJ=1,NN
      KK=1
      FMULTR(JJ,KK)=CR(JJ,KK)
      FMULTI(JJ,KK)=CI(JJ,KK)
142  CONTINUE
      GO TO 143
134  CALL CMMP(PPLUSR,PPLUSI,1,NN,FMULTR,FMULTI,1,CR,CI)
      FIGVGR=CR(1,1)
      FIGVGI=CI(1,1)
143  CONTINUE
      EXPR=COSE(THETA)
      EXPI=SINF(THETA)
      FIVR=FIGVGR*GO(NCVS)
      FIVI=FIGVGI*GO(NCVS)
      FIVER=FREAL(NCVS)*FIVR*EXPR-FIMAG(NCVS)*FIVR*EXPI-FREAL(NCVS)*FIVI
1*EXPI-FIMAG(NCVS)*FIVI*EXPR
      FIVEI=FREAL(NCVS)*FIVR*EXPI+FIMAG(NCVS)*FIVR*EXPR+FREAL(NCVS)*FIVI
1*EXPR-FIMAG(NCVS)*FIVI*EXPI
      GVR=-(RQ(NCVS)*QEXT(NCVS)*FIVER)/(1.0+QEXT(NCVS)/QOEL(NCVS))
      GVI=-(RQ(NCVS)*QEXT(NCVS)*FIVEI)/(1.0+QEXT(NCVS)/QOEL(NCVS))
      PHASE(K)=ATANF(GVI/GVR)
      IF (GVR) 155,156,156
155  PHASE(K)=PHASE(K)+PI
      GO TO 157
156  IF (PHASE(K)) 158,157,157
158  PHASE(K)=PHASE(K)+2.0*PI
157  FONE=FREAL(1)**2+FIMAG(1)**2
      FTWO=FREAL(NCVS)**2+FIMAG(NCVS)**2
      FIGV=FIVR**2+FIVI**2
      GAIN(K)=4.0*RQ(1)*RQ(NCVS)*QEXT(1)*QEXT(NCVS)*FONE*FTWO*FIGV/((1.0
1+QEXT(1)/QOEL(1))**2*(1.0+QEXT(NCVS)/QOEL(NCVS))**2)
      DBGAIN(K)=4.3429*LOGF(GAIN(K))
      IF (WOT) 173,177,173
173  IF (MM) 177,404,177
404  WRITE OUTPUT TAPE 3,52,(Y(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,54,(ZGR(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,55,(ZGI(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,70,FIVR,FIVI,GVR,GVI
177  FRIQ=FRIQ+FFDEL
171  CONTINUE
      IF (SKIP) 101,187,101
187  WRITE OUTPUT TAPE 3,50,(FNU(I),I=1,NCVS)
      WRITE OUTPUT TAPE 3,71,(GAIN(KK),KK=1,NKK)
      WRITE OUTPUT TAPE 3,72,(DBGAIN(KK),KK=1,NKK)
      WRITE OUTPUT TAPE 3,72,(PHASE(KK),KK=1,NKK)
      MM=1
      GO TO 175
C      FORMAT STATEMENTS
      1  FORMAT (54H
      2  FORMAT (2I5,3E8.0,3F7.4)

```

```

3  FORMAT (7F8.4)
4  FORMAT (7F10.2)
5  FORMAT (F10.4)
6  FORMAT (20H NUMBER OF CAVITIES=I3/17H NUMBER OF MODES=I3/12H DC VO
1LTAGE=E10.3)
7  FORMAT (25H      B+                      ,7F12.6)
8  FORMAT (25H      B-                      ,7F12.6)
9  FORMAT (15,2F10.4)
10 FORMAT (/////6H FDEL=F10.4//)
11 FORMAT (///3H M=I5/)
12 FORMAT (25H      DC CAVITY VOLTAGE      ,7E12.3)
13 FORMAT (7H CAVITY26X,1H111X,1H211X,1H311X,1H411X,1H511X,1H611X,1H7
1)
14 FORMAT (12H DC CURRENT=E10.3/30H CENTER FREQUENCY(CYCLES/SEC)=E12.
14/3H A=F7.4/3H B=F7.4/3H C=F7.4////)
15 FORMAT (25H      D                      ,7F12.4)
16 FORMAT (25H      R/O                    ,7F12.2)
17 FORMAT (25H      QO                     ,7F12.2)
18 FORMAT (25H      QEXT                   ,7F12.2)
19 FORMAT (29H CENTER GAP TO CENTER GAP    ,7F12.4)
20 FORMAT (3F10.2)
21 FORMAT (25H      G+                      ,7F12.6)
22 FORMAT (25H      G-                      ,7F12.6)
23 FORMAT (25H      T+                      ,7F12.6)
24 FORMAT (25H      T-                      ,7F12.6)
25 FORMAT (25H      WQ+/WP                 ,7F12.6)
26 FORMAT (25H      RE                     ,7F12.6)
27 FORMAT (25H      WQ-/WP                 ,7F12.6)
28 FORMAT (25H      WQ+/W                  ,7F12.6)
29 FORMAT (25H      WQ-/W                  ,7F12.6)
30 FORMAT (25H      ZETA+                   ,7F12.6)
31 FORMAT (25H      ZETA-                   ,7F12.6)
32 FORMAT (25H      YO+/GO                 ,7F12.6)
33 FORMAT (25H      YO-/GO                 ,7F12.6)
34 FORMAT (25H      E+                      ,7F12.6)
35 FORMAT (25H      E-                      ,7F12.6)
36 FORMAT (25H      F+                      ,7F12.6)
37 FORMAT (25H      F-                      ,7F12.6)
38 FORMAT (25H      C+                      ,7F12.6)
39 FORMAT (25H      C-                      ,7F12.6)
40 FORMAT (25H      M+                      ,7F12.6)
41 FORMAT (25H      M-                      ,7F12.6)
42 FORMAT (25H      H+                      ,7F12.6)
43 FORMAT (25H      H-                      ,7F12.6)
44 FORMAT (25H      GELGO                   ,7F12.6)
45 FORMAT (/////25H      GELGOF            ,7F12.6)
46 FORMAT (25H      QEL                     ,7F12.6)
47 FORMAT (25H      Q                      ,7F12.6)
48 FORMAT (25H      QOEL                   ,7F12.6)
49 FORMAT (7F10.5)
50 FORMAT (//25H      NU                    ,7F12.6)
52 FORMAT (25H      Y                      ,7F12.6)
53 FORMAT (25H      GEL                     ,7F12.6)
54 FORMAT (25H      ZGR                     ,7F12.6)
55 FORMAT (25H      ZGI                     ,7F12.6////)
56 FORMAT (5H NKK=I5/6H FREQ=F10.4/5H DEL=F10.4/)
66 FORMAT (25H      BQPOS                   ,7F12.6)
67 FORMAT (25H      BQMIN                   ,7F12.6)

```

```
70  FORMAT (6H FIVR=F10.6/6H FIVI=F10.6/5H GVR=F10.6/5H GVI=F10.6)
71  FORMAT (//8E14.6/(8E14.6))
72  FORMAT (//8F14.6/(8F14.6))
81  FORMAT(25H      REL.              7F12.6)
    END
```

# POTENTIAL DEPRESSION

```

SUBROUTINE POTDEP(NCVS,VO,FIO,A,B,C,DGAP,VPTR,RSUBD,VBAR,VSUBD)
DIMENSION VOC(10),DGAP(10),VPTR(10),VSUBD(10),VBAR(10),ROEPS(10),O
1DDEV(25)
NCVSP1=NCVS+1
EM=1.759E11
CL=2.998E8
EMC2=EM/CL**2
VOC(NCVSP1)=SQRTF(1.0-1.0/(1.0+EMC2*VO)**2)
ODDEV(1)=2.0
ODDEV(2)=2.0
DO 42 J=4,24,2
ODDEV(J-1)=ODDEV(J-2)+2.0
ODDEV(J)=ODDEV(J-1)
42 CONTINUE
PI=3.14159265
FMGO=376.7
BPC=B**2+C**2
BMC=B**2-C**2
VPTR(NCVSP1)=1.0/(2.0*PI)*FMGO*FIO/VOC(NCVSP1)*(0.5+LOGF(A/B)-BPC/
1(4.0*BMC)+C**4*LOGF(B/C)/BMC**2)
VSUBD(NCVSP1)=VO-VPTR(NCVSP1)/(1.0-VPTR(NCVSP1)/(2.0*VO))
I=0
RD=B
RRD=(B-C)/2.0
GO TO 31
17 RD=RD-RRD
31 I=I+1
TEST=BMC/2.0-BPC/4.0+C**4*LOGF(B/C)/BMC-(B**2-RD**2)/2.0+C**2*LOGF
1(B/RD)
IF (I-1) 10,11,12
10 PAUSE
12 IF (TEST) 13,14,15
13 SIGN2=-1.0
GO TO 16
15 SIGN2=+1.0
16 IF (SIGN+SIGN2) 17,18,17
11 IF (TEST) 19,14,20
19 SIGN=-1.0
GO TO 17
20 SIGN=+1.0
GO TO 17
18 RDD=1.0E-6
IF (RRD-RDD) 14,14,21
21 RD=RD+RRD
RRD=RRD/10.0
GO TO 17
14 RSUBD=RD
DO 22 I=1,NCVS
VOC(I)=SQRTF(1.0-1.0/(1.0+EMC2*VO)**2)
ROEPS(I)=FMGO/PI*FIO/VOC(I)*1.0/(BMC)*(39.37)**2/(1.0-VPTR(I)/(2.0
1*VO))
M=0
27 M=M+1
FM=M
IF (FM/ODDEV(M)-1.0) 40,27,41
41 PAUSE
40 FPD=FM*PI/(2.0*DGAP(I))
IF (C) 28,29,30

```

```

28 PAUSE
30 FI1CD=BESSELF(FPD*C,1,2)
   GO TO 32
29 FI1CD=0.0
32 FK1BD=BESSELF(FPD*B,1,3)
   FI0RD=BESSELF(FPD*RSUBD,0,2)
   FK0RD=BESSELF(FPD*RSUBD,0,3)
   UM=(1.0/FPD)**2*(1.0-FPD*B*FK1BD*FI0RD-FPD*C*FK0RD*FI1CD)/(39.37)*
1*2
   ANR=8.0/(FM**2*PI**2)*ROEPS(I)*SINF(FM*PI/2.0)**2*UM/(1.0-ROEPS(I)
1*UM/(2.0*VO))
   IF (M-1) 23,24,25
23 PAUSE
25 IF (ANR-CHECK) 26,38,38
24 CHE=1.0E-4
   CHECK=ANR*CHE
   SUM=ANR
   GO TO 27
38 SUM=SUM+ANR
   GO TO 27
26 VRAR(I)=VSUBD(NCVSP1)-SUM
22 CONTINUE
   RETURN
   END

```

```

C          SPACE CHARGE REDUCTION FACTOR
SUBROUTINE SPCRD (I,A,B,C,DDG,GP,BP,BE,FK,GG,BB,T)
DIMENSION GP(10),BP(10),BE(10),FK(10)
G=GP(I)
J=0
DG=DDG
20  G=G+DG
    TTT=G**2*(BP(I)**2/((SQRTF(G**2+FK(I)**2)-BE(I))**2)-1.0)
    IF (TTT) 20,20,12
12  TT=SQRTF(TTT)
    J=J+1
    IF (C) 1,3,2
1  PAUSE 11111
2  FJ0TC=BESSELF(TT*C,0,1)
   FJ1TC=BESSELF(TT*C,1,1)
   FJ0GC=BESSELF(G*C,0,2)
   FJ1GC=BESSELF(G*C,1,2)
   FN0TB=BESSELF(TT*B,0,4)
   FN1TB=BESSELF(TT*B,1,4)
   FN0TC=BESSELF(TT*C,0,4)
   FN1TC=BESSELF(TT*C,1,4)
3  FJ0GA=BESSELF(G*A,0,2)
   FJ0GB=BESSELF(G*B,0,2)
   FJ1GB=BESSELF(G*B,1,2)
   FK0GA=BESSELF(G*A,0,3)
   FK0GB=BESSELF(G*B,0,3)
   FK1GB=BESSELF(G*B,1,3)
   FJ0TB=BESSELF(TT*B,0,1)
   FJ1TB=BESSELF(TT*B,1,1)
   BCOTH=(FK0GA*FJ1GB+FK1GB*FJ0GA)/(FK0GB*FJ0GA-FK0GA*FJ0GB)
   IF (C) 4,5,6
4  PAUSE 44444
5  FOFI=TT*B*FJ1TB-G*B*BCOTH*FJ0TB
   GO TO 7
6  FOFI=(TT*B)**2*(FJ1TB*FN1TC-FJ1TC*FN1TB)-TT*B*G*B*BCOTH*(FJ0TB*FN1
1TC-FJ1TC*FN0TB)+TT*G*B**2*FJ1GC/FJ0GC*(FJ1TB*FN0TC-FJ0TC*FN1TB)-(G
2*B)**2*BCOTH*FJ1GC/FJ0GC*(FJ0TB*FN0TC-FJ0TC*FN0TB)
7  IF (J-1) 4,70,69
69 IF (FOFI) 80,10,82
80 SIGN2=-1.0
   GO TO 83
82 SIGN2=+1.0
83 IF (SIGN+SIGN2) 20,8,20
70 IF (FOFI) 64,10,66
64 SIGN=-1.0
   GO TO 20
66 SIGN=+1.0
   GO TO 20
8  DGG=1.0E-6
   IF (DG-DGG) 10,10,9
9  G=G-DG
   DG=DG/10.0
   GO TO 20
10 T=TT
   GG=G
   BB=SQRTF(GG**2+FK(I)**2)
   RETURN
   END

```

```

C          SPACE CHARGE REDUCTION FACTOR
SUBROUTINE SPCRDM(I,A,B,C,DDG,GM,BP,BE,FK,GG,BB,T)
DIMENSION GM(10),BP(10),BE(10),FK(10)
G=GM(I)
J=0
DG=DDG
20  G=G-DG
    TTT=G**2*(BP(I)**2/((SQRTF(G**2+FK(I)**2)-BE(I))**2)-1.0)
    IF (TTT) 20,20,12
12  TT=SQRTF(TTT)
    J=J+1
    IF (C) 1,3,2
1  PAUSE 11111
2  FJ0TC=BESSELF(TT*C,0,1)
   FJ1TC=BESSELF(TT*C,1,1)
   FIOGC=BESSELF(G*C,0,2)
   FI1GC=BESSELF(G*C,1,2)
   FN0TB=BESSELF(TT*B,0,4)
   FN1TB=BESSELF(TT*B,1,4)
   FN0TC=BESSELF(TT*C,0,4)
   FN1TC=BESSELF(TT*C,1,4)
3  FIOGA=BESSELF(G*A,0,2)
   FIOGB=BESSELF(G*B,0,2)
   FI1GB=BESSELF(G*B,1,2)
   FK0GA=BESSELF(G*A,0,3)
   FK0GB=BESSELF(G*B,0,3)
   FK1GB=BESSELF(G*B,1,3)
   FJ0TB=BESSELF(TT*B,0,1)
   FJ1TB=BESSELF(TT*B,1,1)
   BCOTH=(FK0GA*FI1GB+FK1GB*FIOGA)/(FK0GB*FIOGA-FK0GA*FIOGB)
   IF (C) 4,5,6
4  PAUSE 44444
5  FOFI=TT*B*FJ1TB-G*B*BCOTH*FJ0TB
   GO TO 7
6  FOFI=(TT*B)**2*(FJ1TB*FN1TC-FJ1TC*FN1TB)-TT*B*G*B*BCOTH*(FJ0TB*FN1
   1TC-FJ1TC*FN0TB)+TT*G*B**2*FI1GC/FIOGC*(FJ1TB*FN0TC-FJ0TC*FN1TB)-(G
   2*B)**2*BCOTH*FI1GC/FIOGC*(FJ0TB*FN0TC-FJ0TC*FN0TB)
7  IF (J-1) 4,70,69
69 IF (FOFI) 80,10,82
80 SIGN2=-1.0
   GO TO 83
82 SIGN2=+1.0
83 IF (SIGN+SIGN2) 20,8,20
70 IF (FOFI) 64,10,66
64 SIGN=-1.0
   GO TO 20
66 SIGN=+1.0
   GO TO 20
8  DGG=1.0E-6
   IF (DG-DGG) 10,10,9
9  G=G+DG
   DG=DG/10.0
   GO TO 20
10 T=TT
   GG=G
   BB=SQRTF(GG**2+FK(I)**2)
   RETURN
   END

```

```

C          FINDING E,F,C,M,H
SUBROUTINE EFCMH(L,A,B,C,NCVS,VI,X,Y,ZZ,EE,FF,CCK,DUM1,DUM2,DUM3)
DIMENSION X(10),Y(10),ZZ(10),EE(10),FF(10),CCK(10,10),VI(10)
DUM1=DUM2+DUM3
DO 10 I=1,NCVS
  IF (I-1) 15,11,12
15 PAUSE
12 III=I-1
  DO 13 II=1,III
    IF (VI(I)-VI(II)) 13,14,13
13 CONTINUE
11 T=X(I)
  G=Y(I)
  BT=ZZ(I)
  IF (C) 1,2,3
1 PAUSE 11111
3 FN0TB=BESSELF(T*B,0,4)
  FN1TB=BESSELF(T*B,1,4)
  FJ0TC=BESSELF(T*C,0,1)
  FJ1TC=BESSELF(T*C,1,1)
  FN0TC=BESSELF(T*C,0,4)
  FN1TC=BESSELF(T*C,1,4)
  FIOGC=BESSELF(G*C,0,2)
  FI1GC=BESSELF(G*C,1,2)
2 FJ0TB=BESSELF(T*B,0,1)
  FJ1TB=BESSELF(T*B,1,1)
  FIOGA=BESSELF(G*A,0,2)
  FIOGB=BESSELF(G*B,0,2)
  FK0GA=BESSELF(G*A,0,3)
  FK0GB=BESSELF(G*B,0,3)
  FI1GB=BESSELF(G*B,1,2)
  IF (C) 4,5,6
4 PAUSE 44444
5 E=FJ0TB**2+FJ1TB**2
  F=FJ0TB**2*(1.0+G**2/T**2+1.0/((T*B*(FIOGB*FK0GA-FIOGA*FK0GB))**2)
  1)
  CK=2.0*(T*B*FIOGB*FJ1TB+G*B*FI1GB*FJ0TB)/((G**2*B**2+T**2*B**2)*FIOGA*(FJ0TB**2+FJ1TB**2))
  GO TO 7
6 BB=-(T*FJ1TC+G*FI1GC*FJ0TC/FIOGC)/(T*FN1TC+G*FI1GC*FN0TC/FIOGC)
  E=(B**2*((FJ0TB**2+FJ1TB**2)+2.0*BB*(FJ0TB*FN0TB+FJ1TB*FN1TB))+BB**
  12*(FN0TB**2+FN1TB**2))-C**2*((FJ0TC**2+FJ1TC**2)+2.0*BB*(FJ0TC*FN0
  2TC+FJ1TC*FN1TC))+BB**2*(FN0TC**2+FN1TC**2))/((B**2-C**2)
  F=(1.0+G**2/T**2)*(B**2*(FJ0TB+BB*FN0TB)**2-C**2*(FJ0TC+BB*FN0TC)*
  1*2)/((B**2-C**2)+(FJ0TB+BB*FN0TB)**2/(T**2*(B**2-C**2)*(FIOGB*FK0GA
  1-FIOGA*FK0GB)**2)
  CK=2.0*((T*B*FIOGB*(FJ1TB+BB*FN1TB)+G*B*FI1GB*(FJ0TB+BB*FN0TB))-(T
  1*C*FIOGC*(FJ1TC+BB*FN1TC)+G*C*FI1GC*(FJ0TC+BB*FN0TC)))/(E*FIOGA*(B
  2**2-C**2)*(G**2+T**2))
7 EE(I)=E
  FF(I)=F
  CCK(L,I)=CK
  GO TO 10
14 EE(I)=EE(II)
  FF(I)=FF(II)
  CCK(L,I)=CCK(L,II)
10 CONTINUE
  RETURN
  END

```

```

C          FREQUENCY FUNCTION SUBROUTINE
SUBROUTINE FREQ(Q,TU,DELO,NCVS ,FREAL,FIMAG)
DIMENSION Q(10),TU(10),FREAL(10),FIMAG(10)
DO 10 I=1,NCVS
CFCTR = Q(I)*(2.+DELO+TU(I))*(DELO-TU(I))/((1.+DELO)*(1.+TU(I)))
FREAL(I) = 1./(1.+CFCTR**2)
10 FIMAG(I) = -CFCTR*FREAL(I)
RETURN
END

```

```

C          INTERPOLATION
SUBROUTINE TERPOL(N,L,Y,X,ANSR)
DIMENSION Y(3,10,10),ANSR(10)
DO 10 I=1,N
A=Y(2,L,I)
B=(Y(3,L,I)-Y(1,L,I))/0.16
C=(Y(1,L,I)+Y(3,L,I)-2.0*Y(2,L,I))/0.0128
ANSR(I)=A+B*X+C*X**2
10 CONTINUE
RETURN
END

```

```

C          AGAIN INTERPOLATION
SUBROUTINE AINTER(N,L,Y,X,YY)
DIMENSION Y(3,10,10),YY(10,10)
DO 10 I=1,N
A=Y(2,L,I)
B=(Y(3,L,I)-Y(1,L,I))/0.16
C=(Y(1,L,I)+Y(3,L,I)-2.0*Y(2,L,I))/0.0128
YY(L,I)=A+B*X+C*X**2
10 CONTINUE
RETURN
END

```

C

## FORMATION OF MATRICES

```

SUBROUTINE FOMATR(N,I,NCVS,CKP,CKM,FMP,FMM,FHP,FHM,ZETAP,ZETAM,YOG
10P,YOGOM,PPLUSR,PPLUSI,COLMR,COLMI,FLMPR,FLMPI)
  DIMENSION CKP(10,10),CKM(10,10),FMP(10,10),FMM(10,10),FHP(10,10),F
1HM(10,10),ZETAP(10,10),ZETAM(10,10),YOGOP(10,10),YOGOM(10,10),PPLU
2SR(20,20),PPLUSI(20,20),COLMR(20,20),COLMI(20,20),FLMPR(20,20),FLM
3PI(20,20)
  L=0
  NN=2*N
  DO 1 K=1,NN,2
    J=1
    L=L+1
    PPLUSR(J,K)=CKP(L,I)*FMP(L,I)*FHP(L,I)*YOGOP(L,I)/(1.0+ZETAP(L,I))
    PPLUSR(J,K+1)=-CKM(L,I)*FMM(L,I)*FHM(L,I)*YOGOM(L,I)/(1.0-ZETAM(L,
1 I))
    PPLUSI(J,K)=0.0
    PPLUSI(J,K+1)=0.0
1  CONTINUE
  L=0
  DO 2 J=1,NN,2
    K=1
    L=L+1
    COLMR(J,K)=CKP(L,I)*FMP(L,I)/(2.0*(1.0+ZETAP(L,I)))
    COLMR(J+1,K)=CKM(L,I)*FMM(L,I)/(2.0*(1.0-ZETAM(L,I)))
    COLMI(J,K)=0.0
    COLMI(J+1,K)=0.0
2  CONTINUE
  DO 3 J=1,NN
    DO 3 K=1,NN
      L=1
      FLMPR(J,K)=COLMR(J,L)*PPLUSR(L,K)
      FLMPI(J,K)=COLMI(J,L)*PPLUSI(L,K)
3  CONTINUE
  RETURN
END

```

C

## FORMATION OF D MATRIX

```

SUBROUTINE DMATRIX(NCVSP1,N,I,BQPOS,BQMIN,BE,S,DR,DI,FIR,FII)
  DIMENSION BQPOS(10,10),BQMIN(10,10),BE(10),S(10),DR(20,20),DI(20,2
10),FIR(20,20),FII(20,20)
  DO 1 J=1,N,2
    M=0
    DO 2 K=1,N,2
      M=M+1
      IF (J-K) 3,4,3
3    DR(J,K)=0.0
      DI(J,K)=0.0
      DR(J+1,K)=0.0
      DI(J+1,K)=0.0
      DR(J,K+1)=0.0
      DI(J,K+1)=0.0
      DR(J+1,K+1)=0.0
      DI(J+1,K+1)=0.0
      FIR(J,K)=0.0
      FII(J,K)=0.0
      FIR(J+1,K)=0.0
      FII(J+1,K)=0.0
      FIR(J,K+1)=0.0
      FII(J,K+1)=0.0
      FIR(J+1,K+1)=0.0
      FII(J+1,K+1)=0.0
      GO TO 2
4    SXP=SINF(BE(NCVSP1)*S(I))
      CXP=COSF(BE(NCVSP1)*S(I))
      SYP=SINF(BQPOS(M,NCVSP1)*S(I))
      CYP=COSF(BQPOS(M,NCVSP1)*S(I))
      SYM=SINF(BQMIN(M,NCVSP1)*S(I))
      CYM=COSF(BQMIN(M,NCVSP1)*S(I))
      DR(J,K)=SXP*SYP+CXP*CYP
      DI(J,K)=SYP*CXP-SXP*CYP
      DR(J+1,K+1)=CXP*CYM-SXP*SYM
13    DI(J+1,K+1)=-(SXP*CYM+SYM*CXP)
      FIR(J,K)=1.0
      FII(J,K)=0.0
      FIR(J+1,K+1)=1.0
      FII(J+1,K+1)=0.0
2    CONTINUE
1    CONTINUE
  RETURN
END

```

```

C          COMPLEX MATRIX MULTIPLICATION (UP TO 10X10)
C          AXB=C
C          N = ROWS OF A AND C
C          M = ROWS OF B AND COLUMNS OF A
C          L = COLUMNS OF B AND C
SUBROUTINE CMMP(AR,AI,N,M,BR,BI,L,CR,CI)
DIMENSION AR(20,20),AI(20,20),BR(20,20),BI(20,20),CR(20,20),CI(20,
120),D(20,20),E(20,20)
CALL RMMP(AR,N,M,BR,L,D)
CALL RMMP(AI,N,M,BI,L,E)
DO 10 I=1,N
DO 10 J=1,L
10 CR(I,J)=D(I,J)-E(I,J)
CALL RMMP(AI,N,M,BR,L,D)
CALL RMMP(AR,N,M,BI,L,E)
DO 20 I=1,N
DO 20 J=1,L
20 CI(I,J)=D(I,J)+E(I,J)
RETURN
END

```

```

C          COMPLEX MATRIX SUBTRACTION
SUBROUTINE CMSUBT(ZGGL,FI,NN,SUM)
DIMENSION ZGGL(20,20),FI(20,20),SUM(20,20)
DO 1 J=1,NN
DO 1 K=1,NN
SUM(J,K)=FI(J,K)-ZGGL(J,K)
1 CONTINUE
RETURN
END

```

```

C          REAL MATRIX MULTIPLICATION
SUBROUTINE RMMP (A,N,M,B,L,C)
DIMENSION A(20,20),B(20,20),C(20,20)
DO 5 I=1,N
DO 5 J=1,L
C(I,J)=0.0
DO 5 K=1,M
5 C(I,J)=C(I,J)+A(I,K)*B(K,J)
RETURN
END

```

## REFERENCES

1. R. L. Pease, "A More Precise Analysis of Multicavity Klystrons I, " Technical Report No. 264 [U], Lincoln Laboratory, M.I. T. (28 March 1962). DDC No. 275469.
2. \_\_\_\_\_, "A More Precise Analysis of Multicavity Klystrons II, " To be published.
3. \_\_\_\_\_, "Calculation of Potential Depression in Klystron Gaps, " Technical Report No. 316 [U], Lincoln Laboratory, M.I. T. (23 May 1963).

**UNCLASSIFIED**

**UNCLASSIFIED**